12. ONLINE SUPPLEMENT FOR BASIC CONCEPTS IN POPULATION PHARMACOKINETIC MODELING

This online supplement provides additional commentary and example code for the paper "Basic Concepts In Population Modeling, Simulation, And Model-Based Drug Development - Part 2: Introduction To Population Pharmacokinetic Modeling Methods" by Mould and Upton. Code is provided for several examples in a format for the widely used R data analysis and statistical language [1]. This can be saved by cutting and pasting into a text-editor (preferably with syntax highlighting for R) and then executing in R. No additional R packages are required for the supplied code.

Example data and example control streams are provided for NONMEM for some representative, didactic models [2]. Complete models and data are provided for the examples so that the reader is provided with a working example of the model in question. Please note that these models are not the best possible model for the supplied data, but are intended to demonstrate the principles of coding the models. The example data should be cut and pasted into a spreadsheet program and saved as a comma separated file with the name indicated. The control streams can be cut and pasted into a text editor then saved into the same folder as the data file for executing the NONMEM run. Models were developed using NONMEM Version 7 level 2.0. Wings for NONMEM Version 720 [3] was used a "front end" for the NONMEM program.

12.1 Concept of Censoring Due to the LLOQ

In a key paper [4], Beal discussed 7 potential methods for the population pharmacokinetic modeling of data with BLQ observations. These included: Method 1 (M1): Not including the BLQ observations (implemented as MDV=1 in NONMEM). Method 2 (M2): Weighting the likelihood of model predicted concentrations that are not BLQ with the probability that they are above the LLOQ (YLO in NONMEM). Method 3 (M3): Including the probability that the model predicted concentrations for BLQ observations are below the LLOQ as part of the maximum likelihood estimation (F_FLAG in NONMEM).

The effect of LLOQ censoring and the contribution of the M3 method are illustrated in using a simulated data set as shown in Figure 1. It can be seen that the M3 method was

not needed when there was extensive data from higher doses (200 & 50 ug doses, all LLOQ values) that produced full time-courses of uncensored data for some subjects. The M3 method was able to correct for censored data when censoring was present and some subjects had partial data (50 µg dose only and LLOQ = 0.05). The M3 method was not able to correct for censoring when data informing a second compartment were completely missing (50 µg dose only and LLOQ = 0.1). The M3 and YLO methods may have penalties with respect to model run-time and stability that may preclude their use in some circumstances. Note also that the M3 and YLO methods are for estimation, not simulation.

Figure 1 – The effect of censoring due to LLOQ

A data set was simulated for a study where fentanyl was given in 2 doses (50 and 200 ug) via two routes (i.v. and s.c.) to 20 subjects (5 per group) [6, 6]. A two compartment model with (population predictions, black line) and without (population predictions, red line) the M3 method for censored data was fitted to simulated data where the LLOQ was set at 3 different concentrations (0, 0.05 and 0.1 ng/ml) as described above. Furthermore, models were fitted to the data for all doses and routes subjects (20 subjects, upper row) and to data for the 50 µg s.c. doses (5 subjects, lower row). As the lowest observed concentrations were for the 50 µg s.c. cohort, the data for this cohort were most affected by censoring and the fitted results are shown only for this cohort.

The upper row shows a scenario where there higher doses without censoring can inform the model even with substantial censoring of the lower dose data. In this case, models using M1 provided similar fits to models using M3 regardless of the LLOQ value. The lower row shows a scenario where the higher dose data were not used for model building. When the LLOQ was 0.05, the M3 method was able to provide a similar fit than that for the uncensored data, whereas not using M3 introduced a bias into the fit. When the LLOQ was 0.1, the M3 method was not able to "rescue" the situation, as all subjects were missing the data informing the second compartment of the model. Example datasets and control streams showing the use of the YLO and M3 methods in NONMEM are also shown below.

Fitting 50 & 200 ug doses and s.c. & i.v. routes
Results for 50 ug dose - s.c. route
Red = without M3 method, Black = with M3 method



Source: P:/NONMEM102/LLOQmethodsAllData LLOQtest_Conc_vs_time_M3_Alldata 2012-12-06 16:36:51

Fitting 50 ug dose - s.c. route only
Results for 50 ug dose - s.c. route
Red = without M3 method, Black = with M3 method



Source: P:/NONMEM102/LLOQmethodsPartialData LLOQtest_Conc_vs_time_M3_PartialDataset 2012-12-06 16:38:08

12.1.1 Example data for the YLO method

Filename should be 2comp_YOdata.csv

CID	TIME	AMT	DV	MDV
1	0	50	.	1
1	2.5	.	0.41602	0
1	5	.	0.40408	0
1	10	.	0.32019	0
1	15	.	0.36337	0
1	30	.	0.091612	0
1	60	.	0.072907	0
1	90	.	0.056053	0
1	120	.	0.065449	0
1	180	.	0.053612	0
1	240	.	.	1
1	300	.	0.052835	0
1	360	.	0.061585	0
1	480	.	.	1
1	600	.	.	1
2	0	50	.	1
2	2.5	.	0.22745	0
2	5	.	0.48887	0
2	10	.	0.34948	0
2	15	.	0.19076	0
2	30	.	0.17412	0
2	60	.	0.11533	0
2	90	.	0.10667	0
2	120	.	0.088994	0
2	180	.	0.068695	0
2	240	.	.	1
2	300	.	0.065073	0
2	360	.	0.078097	0
2	480	.	0.066583	0
2	600	.	0.051223	0

12.1.2 Example NONMEM control stream for the YLO method

```
$PROBLEM TWO COMPARTMENT FIRST ORDER ABSORPTION WITH YLO (BEAL METHOD 2)

$INPUT ID TIME AMT DV MDV

$DATA   2comp_YLOdata.CSV IGNORE=C

$SUBROUTINES  ADVAN4 TRANS4

$PK
   CL = THETA(1)*EXP(ETA(1))
   V2 = THETA(2)
Q  = THETA(3)
   V3 = THETA(4)

   KA = THETA(5)

   S2 = V2

$ERROR  ; Beal Method 2
;residual error is coded here using THETA's rather than EPS's
  ADD = THETA(6)
  PROP = F*THETA(7)
  SD = SQRT(ADD*ADD+PROP*PROP) ; combined error model

  Y = F + SD*EPS(1)
  IPRED=F

  YLO=0.05 ;LLOQ
  PRB=PR_Y  ; probability that F is less than LLOQ

$THETA
 (0,0.6,)   ;CLPOP
 (0,40,)    ;V2POP
 (0,3,)     ;QPOP
 (0,130,)   ;V3POP
 (0,0.02,)  ;KAPOP

 (0,0.002,) FIX ;RUVSDADD   not estimated for example dataset
 (0,0.2,)   ;RUVSDPROP

$OMEGA
0.2 ;CLPPV

$SIGMA 1 FIX

;$SIMULATE(1234567) SUBPROBLEMS=1

$ESTIMATION METHOD=1 LAPLACIAN INTERACTION NUMERICAL SLOW MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9 PRINT=1

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID TIME  AMT  MDV PRB CWRES IPRED NOPRINT ONEHEADER FILE=*.fit
;PRB is probability that concentration is below the LLOQ
```

12.1.3 Example data for the M3 method

Filename should be 2comp_M3data.csv

CID	TIME	AMT	DV	MDV	BLQ
1	0	50	.	1	0
1	2.5	.	0.41602	0	0
1	5	.	0.40408	0	0
1	10	.	0.32019	0	0
1	15	.	0.36337	0	0
1	30	.	0.091612	0	0
1	60	.	0.072907	0	0
1	90	.	0.056053	0	0
1	120	.	0.065449	0	0
1	180	.	0.053612	0	0
1	240	.	.	1	1
1	300	.	0.052835	0	0
1	360	.	0.061585	0	0
1	480	.	.	1	1
1	600	.	.	1	1
2	0	50	.	1	0
2	2.5	.	0.22745	0	0
2	5	.	0.48887	0	0
2	10	.	0.34948	0	0
2	15	.	0.19076	0	0
2	30	.	0.17412	0	0
2	60	.	0.11533	0	0
2	90	.	0.10667	0	0
2	120	.	0.088994	0	0
2	180	.	0.068695	0	0
2	240	.	.	1	1
2	300	.	0.065073	0	0
2	360	.	0.078097	0	0
2	480	.	0.066583	0	0
2	600	.	0.051223	0	0

12.1.4 Example NONMEM control stream for the M3 method

```
$PROBLEM TWO COMPARTMENT FIRST ORDER ABSORPTION WITH M3 (BEAL METHOD 3)

$INPUT ID TIME AMT DV MDV BLQ
;BLQ is 0 for observed concentrations, 1 for missing concentrations

$DATA   2comp_M3data.csv IGNORE=C

$SUBROUTINES  ADVAN4 TRANS4

$PK
   CL = THETA(1)*EXP(ETA(1))
   V2 = THETA(2)
Q  = THETA(3)
   V3 = THETA(4)

   KA = THETA(5)

   S2=V2

$ERROR  ; Beal Method 3
;residual error is coded here using THETA's rather than EPS's
  ADD = THETA(6)
  PROP = F*THETA(7)
  SD = SQRT(ADD*ADD+PROP*PROP) ;combined error model
  LLOQ=0.05                    ;lower limit of quantification

  IF (BLQ.EQ.0) THEN
   F_FLAG=0                    ;regular likelihood for measured concentrations
   Y = F + SD*EPS(1)
  ENDIF

  IF (BLQ.EQ.1) THEN
   F_FLAG=1                    ;probability that F is less than LLOQ for missing concentrations
   Y=PHI((LLOQ-F)/SD)
  ENDIF

  IPRED=F

$THETA
 (0,0.6,)   ;CLPOP
 (0,40,)    ;V2POP
 (0,3,)     ;QPOP
 (0,130,)   ;V3POP
 (0,0.02,)  ;KAPOP

 (0,0.002,) FIX ;RUVSDADD  not estimated for example dataset
 (0,0.2,)       ;RUVSDPROP

$OMEGA
0.2 ;CLPPV

$SIGMA 1 FIX

;$SIMULATE(1234567) SUBPROBLEMS=1
```

```
$ESTIMATION METHOD=1 LAPLACIAN INTERACTION NUMERICAL SLOW MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9 PRINT=1

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID TIME  AMT  MDV BLQ CWRES IPRED NOPRINT ONEHEADER FILE=*.fit
;PRED is population predicted concentration when BLQ=0; PRED is probability that missing concentration is below LLOQ when BLQ=1
;IPRED is individual predicted concentration for both BLQ=0 and BLQ=1
```

12.2 Concepts of Marginal Maximum Likelihood

The concepts of a marginal maximum likelihood can be illustrated graphically by considering one subject from a study population. Figure 2 shows an example for a 1 compartment, first order absorption pharmacokinetic model (100% bioavailability). Clearance (CL) was log-normally distributed ($CL = CL_{pop}*exp(ETA)$) and was the only population parameter. Top panel: The observed concentrations (C_{obs}, symbols) and the predicted concentrations (solid line) for the subject given their true clearance value. The predicted concentrations for when clearance is CL_{pop} (dashed line) show this subject had slightly lower clearance (higher concentrations) than population values due to between subject variability (BSV) in clearance. Middle panel: The maximum likelihood was calculated based on C_{obs} and model predicted concentrations ($\hat{C}$) calculated using various values of CL_{pop} and ETA. The likelihood profile is shown as a contour plot - there is a "plane of low likelihood" and a tall, thin peak of higher likelihood with a maximum value for this subject when CL_{pop} is 2 and ETA is approximately -0.13. Lower panel: The marginal likelihood for an ETA value is the area under the curve of the likelihood at a particular value of CL_{pop}[7]. Three values of CL_{pop} are shown, which transect the likelihood "peak" at different points. The combination of CL_{pop} and ETA with the highest AUC gives the maximum likelihood estimate of their values given the data. In this example, the shape of the likelihood surface has been visualized by calculating it for many combinations of CL_{pop} and ETA. This "grid search" algorithm is prohibitively slow if implemented in population modeling software. The task of the estimation method is to find the marginal likelihoods and parameter values with speed and accuracy. See Section 12.2.1 for examples of likelihood objective functions in the R language.

Figure 2 - Marginal likelihood in mixed effect models



12.2.1 Likelihood Objective Functions in R

While the mathematics and literature on likelihood objective functions are complex, in programming languages such as R (www.r-project.org) the functions are considerably simpler as they can make use of the dnorm function for calculating distribution density values [8]. Some representative likelihood objective functions coded in R are given below. The code will be easier to read in a text editor that allows syntax highlighting for R. The R code can then be cut and pasted into an R console. It's most informative to do this section by section so that the results of each line of code can be inspected. Try changing parameter values and see what happens to the OFV value.

```
#USING R CODE TO EXPLORE LIKELIHOOD BASED OBJECTIVE FUNCTIONS
#See for www.r-project.org for more information about R

#-------------------------------------------------------------------------------
#Make some observed data
  time <- c(1,2,3,4,6,8,12,18,24)

dose<- 100
CLtrue<- 2
Vtrue<- 10
KAtrue<- 0.5

  #One compartment first order absorption
  Cobs <- dose/Vtrue*(KAtrue/(KAtrue-CLtrue/Vtrue))*(exp(-CLtrue/Vtrue*time)-exp(-KAtrue*time))

set.seed(123)
  Cobs <- Cobs+rnorm(n=length(time), mean=0, sd=Cobs*0.1)  #Add proportional residual error of 10%
plot(Cobs ~ time)


#-------------------------------------------------------------------------------
#Using the dnorm function to calculate distribution densities for a point on a normal distribution

   #For a normal distribution with mean =2 and standard deviation of 1
   #2 has the highest distribution density value - it's in the center of the normal distribution
density<- dnorm(2, mean=2, sd=1)
density

   #1 is less likely to be in the distribution, it's to the left of the mean
density<- dnorm(1, mean=2, sd=1)
density

   #-1 is even less likely, it's further to the left of the mean
density<- dnorm(-1, mean=2, sd=1)
density

   #3 is less likely to be in the distribution, it's to the right of the mean
density<- dnorm(3, mean=2, sd=1)
density

   #5 is even less likely, it's further to the right of the mean
density<- dnorm(5, mean=2, sd=1)
density

   #dnorm does return a normal distribution!
    x <- seq(from=-1, to=5, by=0.1)
densities<- dnorm(x, mean=2, sd=1)
plot(densities ~ x, type="l")


#-------------------------------------------------------------------------------
#FIXED-EFFECT LIKELIHOOD OBJECTIVE FUNCTION

#Estimated parameters are CL, V, KA and SIGMA
dose<- 100
   CL <- 2
   V <- 10
   KA <- 0.5
```

```
#Model predicted concentrations given these parameter values
   Chat <- dose/V*(KA/(KA-CL/V))*(exp(-CL/V*time)-exp(-KA*time))
plot(Cobs ~ time)
points(Chat ~ time, type="l")

#Calculate likelihood
   #Proportional residual error of 10%
    SIGMA <- Chat*0.10
   #The log of the distribution density values for Cobs for a distribution with mean Chat and
standard deviation of SIGMA
densities  <- dnorm(Cobs,mean=Chat,sd=SIGMA,log=T)
   #Calculate the negative loglikelihood for all data points by summing in the log domain
(multiplication of likelihoods)
negloglike<- -1*sum(densities)

#Objective function value for this observed data and these parameters
#Optimisation is used to find the parameters and sigma that give the lowest OFV
    OFV <- negloglike
    OFV


#-------------------------------------------------------------------------------
#MIXED EFFECT LIKELIHOOD OBJECTIVE FUNCTION
#Example is for one subject in the population (other subjects are handled in the same way)

#Estimated parameters are CLpop, OMEGA, V, KA and SIGMA
dose<- 100
CLpop<- 2   #Population value of clearance
eta<- -0.5  #eta reflects difference between CL for this subject and CLpop
   V <- 10
   KA <- 0.5

 #Model predicted concentrations given these parameter values
   #CL is log-normally distributed in the population
    CL <- CLpop*exp(eta)
    Chat <- dose/V*(KA/(KA-CL/V))*(exp(-CL/V*time)-exp(-KA*time))
plot(Cobs ~ time)
points(Chat ~ time, type="l")

#Calculate likelihood
   #Proportional residual error of 10%
    SIGMA <- Chat*0.10
   #The log of the distribution density values for Cobs for a distribution with mean Chat and
standard deviation of SIGMA
densitiesSigma<- dnorm(Cobs,mean=Chat,sd=SIGMA,log=T)

   #Between subject variability of 20% for CL
    OMEGA <- 0.2
   #The log of the distribution density values for 0 being in a distribution with mean eta and
standard deviation of OMEGA
   #Expected value of eta is zero
densitiesOmega<- dnorm(0,eta,OMEGA,log=T)

   #Calculate the negative loglikelihood for this subject by summing in the log domain
(multiplication of likelihoods)
   #Other subjects are also include in this summation (not shown here)
negloglike<- -1*sum(densitiesSigma,densitiesOmega)

#Objective function value for this observed data and these parameters
#Optimisation is used to find the parameters omega and sigma that give the lowest OFV
   OFV <- negloglike
   OFV


#-------------------------------------------------------------------------------
#BAYES LIKELIHOOD OBJECTIVE FUNCTION
#Example is for one subject

#Estimated parameters are CLi, V and KA
dose<- 100
CLpop<- 2  #Population value of CL

CLi<- 1.75  #Individual value of CL in this subject
   V <- 10
   KA <- 0.5

#Model predicted concentrations given these parameter values
  Chat <- dose/V*(KA/(KA-CLi/V))*(exp(-CLi/V*time)-exp(-KA*time))
plot(Cobs ~ time)
points(Chat ~ time, type="l")

#Calculate likelihood
    #Posterior component
    #Proportional residual error of 10%
sigma<- Chat*0.1
```

```
    #The log of the distribution density values for Cobs for a distribution with mean Chat and
standard deviation of sigma
densitiesPosterior<- dnorm(Cobs, Chat, sigma, log=T)

#Prior component
omega<- 0.2
    #For a log-normal distribution of CLi (CLi = CLpop*exp(eta))
eta<- log(CLi/CLpop)
    #The log of the distribution density values for 0 being in a distribution with mean eta and
standard deviation of omega
densitiesPrior<- dnorm(0, eta, omega, log=T)

#Calculate negative log likelihood
    #densitiesPosterior penalizes if the fit of the data is a poor
    #densitiesPrior penalizes if CLi is too different from CLpop
negloglike<- -1*sum(densitiesPosterior,densitiesPrior)

#Objective function value for this observed data and these parameters
#Optimisation is used to find the parameters that give the lowest OFV
      OFV <- negloglike
      OFV
```

12.3 Example NONMEM Code for some Structural Models

12.3.1 Data for structural model examples

File name should be 2comp_iv_data.csv

CID	TIME	AMT	DV	MDV	WT
1	0	50	.	1	70
1	2.5	.	0.26334	0	70
1	5	.	0.33196	0	70
1	10	.	0.28416	0	70
1	15	.	0.21946	0	70
1	30	.	0.116114	0	70
1	60	.	0.076738	0	70
1	90	.	0.061304	0	70
1	120	.	0.058616	0	70
1	180	.	0.053336	0	70
1	240	.	0.05046	0	70
1	300	.	0.034764	0	70
1	360	.	0.03508	0	70
1	480	.	0.038228	0	70
1	600	.	0.020326	0	70
2	0	100	.	1	60
2	2.5	.	2.1238	0	60
2	5	.	1.88695	0	60
2	10	.	1.12165	0	60
2	15	.	0.7418	0	60
2	30	.	0.56585	0	60
2	60	.	0.268845	0	60
2	90	.	0.418505	0	60
2	120	.	0.211785	0	60
2	180	.	0.238055	0	60
2	240	.	0.270265	0	60
2	300	.	0.112965	0	60
2	360	.	0.229935	0	60
2	480	.	0.17821	0	60
2	600	.	0.13506	0	60
3	0	200	.	1	65
3	2.5	.	5.6644	0	65
3	5	.	6.1286	0	65
3	10	.	3.4058	0	65
3	15	.	2.1382	0	65
3	30	.	1.1687	0	65
3	60	.	0.75104	0	65
3	90	.	0.61568	0	65

3	120	.	0.66896	0	65
3	180	.	0.69688	0	65
3	240	.	0.61602	0	65
3	300	.	0.53642	0	65
3	360	.	0.46574	0	65
3	480	.	0.43812	0	65
3	600	.	0.27616	0	65

12.3.2 NONMEM Code - 2 compartment model

```
$PROBLEM - 2 COMP SYSTEMIC MODEL IV DOSE

$INPUT ID TIME AMT DV MDV WT

$DATA 2comp_iv_data.csv IGNORE=C

$SUBROUTINES  ADVAN3 TRANS4

$PK
   CL = THETA(1)*EXP(ETA(1))
   V1 = THETA(2)
   Q = THETA(3)
   V2 = THETA(4)

   S1 = V1

$THETA
 (0,1,)    ;POPCL
 (0,40,)   ;POPV1
 (0,3,)    ;POPQ
 (0,100,)  ;POPV2

$OMEGA
0.09 ;CLPPV

$SIGMA
0.04 ;RUVPROP

$ERROR
    Y = F*(1+ERR(1))
    IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID CL MDV WT CWRES IPRE  NOPRINT ONEHEADER FILE=*.fit
```

12.3.3 NONMEM Code - 2 compartment model with allometric scaling

```
$PROBLEM - 2 COMP SYSTEMIC MODEL IV DOSE WITH ALLOMETRIC SCALING

$INPUT ID TIME AMT DV MDV WT
;WT is body weight for each subject

$DATA 2comp_iv_data.csv IGNORE=C

$SUBROUTINES  ADVAN3 TRANS4

$PK
   WTS = 70  ;standard weight
```

```
   CL = THETA(1)*(WT/WTS)**0.75*EXP(ETA(1))
   V1 = THETA(2)*(WT/WTS)**1
   Q = THETA(3)*(WT/WTS)**0.75
   V2 = THETA(4)*(WT/WTS)**1

   S1 = V1

$THETA
 (0,1,)    ;POPCL
 (0,40,)   ;POPV1
 (0,3,)    ;POPQ
 (0,100,)  ;POPV2

$OMEGA
0.09 ;CLPPV

$SIGMA
0.04 ;RUVPROP

$ERROR
    Y = F*(1+ERR(1))
    IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID CL MDV WT CWRES IPRE  NOPRINT ONEHEADER FILE=*.fit
```

12.3.4 NONMEM Code - 2 compartment model via differential equations

```
$PROBLEM - 2 COMP SYSTEMIC MODEL IV DOSE VIA DIFFERENTIAL EQUATIONS

$INPUT ID TIME AMT DV MDV WT

$DATA 2comp_iv_data.csv IGNORE=C

$SUBROUTINES  ADVAN13 TOL=9

$MODEL
COMP=(CENT, DEFDOS, DEFOBS)
COMP=(PERIPH)

$PK
   CL = THETA(1)*EXP(ETA(1))
   V1 = THETA(2)
Q  = THETA(3)
   V2 = THETA(4)

   S1 = V1

$DES
    C1 =  A(1)/V1 ; Central compartment concentration
    C2 =  A(2)/V2 ; Peripheral compartment concentration
```

```
DADT(1) =  Q*C2 -Q*C1 -CL*C1
DADT(2) =  Q*C1 -Q*C2
;left and right hand sides of these equations are rates (mass/time)
;mass/time can be either concentration*flow or amount*rate constant

$THETA
 (0,1,)    ;POPCL
 (0,40,)   ;POPV1
 (0,3,)    ;POPQ
 (0,100,)  ;POPV2

$OMEGA
0.09 ;CLPPV

$SIGMA
0.04 ;RUVPROP

$ERROR
    A1=A(1)  ;amount in first compartment
    A2=A(2)  ;amount in second compartment

    Y = F*(1+ERR(1))
    IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID CL A1 A2 MDV WT CWRES IPRE  NOPRINT ONEHEADER FILE=*.fit
```

12.3.5 NONMEM Code - 2 compartment model with saturable elimination

```
$PROBLEM - 2 COMP SYSTEMIC MODEL IV DOSE WITH SATURABLE ELIMINATION

$INPUT ID TIME AMT DV MDV WT

$DATA 2comp_iv_data.csv IGNORE=C

$SUBROUTINES  ADVAN13 TOL=9

$MODEL
COMP=(CENT, DEFDOS, DEFOBS)
COMP=(PERIPH)

$PK
   VMAX = THETA(1)*EXP(ETA(1))
   KM = THETA(2)
   V1 = THETA(3)
Q  = THETA(4)
   V2 = THETA(5)

   S1 = V1

$DES
    C1 =  A(1)/V1 ; Central compartment concentration
```

```
    C2 =  A(2)/V2 ; Peripheral compartment concentration

DADT(1) =  Q*C2 -Q*C1 -(VMAX*C1/(KM+C1))
DADT(2) =  Q*C1 -Q*C2

$THETA
 (0,10,)   ;POPVMAX
 (0,0.5,)  ;POPKM
 (0,40,)   ;POPV1
 (0,3,)    ;POPQ
 (0,100,)  ;POPV2

$OMEGA
 0.09      ;VMAXPPV

$SIGMA
0.04 ;RUVPROP

$ERROR
    A1=A(1)  ;table amount in first compartment
    A2=A(2)  ;table amount in second compartment

    Y = F*(1+ERR(1))
    IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID VMAX KM A1 A2 MDV WT CWRES IPRE  NOPRINT ONEHEADER FILE=*.fit
```

12.4 The relationship between rate constants, volumes and clearance and half-lives

The equations for the three compartment model are taken from [9].

```
#Equations to convert between rate constants, volumes and clearances and half-lives for 1, 2 and 3 compartment mammillary pharmacokinetic models

#-----------------------------------------------------------------------
#One compartment model
#Hydraulic parameterization
CL <- 2
V  <- 10

#Rate constant parameterization
k10 <- CL/V
V <- V

#Half-life
thalf<- log(2)/k10  #natural logarithm


#-----------------------------------------------------------------------
#Two compartment model
#Hydraulic parameterization
CL <- 2
V1  <- 10
Q <- 0.5
V2  <- 30

#Rate constant parameterization
k10 <- CL/V1
k12 <- Q/V1
k21 <- k12*V1/V2

#Half-lives
beta<- 0.5*(k12+k21+k10-sqrt((k12+k21+k10)^2-4*k21*k10))
alpha<- k21*k10/beta

thalf_alpha<- log(2)/alpha
thalf_beta<- log(2)/beta


#-----------------------------------------------------------------------
#Three compartment model
#Hydraulic parameterization
CL <- 2
V1  <- 10
Q12 <- 0.5
V2  <- 30
Q13 <- 0.3
V3  <- 40

#Rate constant parameterization
k10 <- CL/V1
k12 <- Q12/V1
k21 <- k12*V1/V2
```

```
k13 <- Q13/V1
k31 <- k13*V1/V3

#Half-lives
j <- k12+k10+k21+k31+k13
k <- k12*k31+k10*k21+k10*k31+k21*k31+k13*k21
l <- k10*k21*k31

m <- (3*k - j^2)/3
n <- (2*j^3 - 9*j*k + 27*l)/27
Q <- (n^2)/4 + (m^3)/27

alpha<- sqrt(-1*Q)
beta<- -1*n/2
rho<- sqrt(beta^2+alpha^2)
theta<- atan2(alpha,beta)

lambda1 <- j/3 + rho^(1/3)*(cos(theta/3) + sqrt(3)*sin(theta/3))
lambda2 <- j/3 + rho^(1/3)*(cos(theta/3) - sqrt(3)*sin(theta/3))
lambda3 <- j/3 -(2*rho^(1/3)*cos(theta/3))

thalf_alpha<- log(2)/lambda1
thalf_beta<- log(2)/lambda2
thalf_gamma<- log(2)/lambda3
```

12.5 Compartmental models as hydraulic analogues

Visualizing compartmental models as hydraulic analogues can be a useful conceptual tool. Distribution volumes can be represented as tanks; the bigger the distribution volume, the bigger the tank. Drug concentration in a tank is represented by the height of water in a tank. A dose is represented by adding water to the "central" tank. An intravenous infusion may be considered a constant flow of water coming out of a tap; conversely a bolus is quickly adding a fixed volume of water to the tank. Elimination is represented by a tap at the bottom of the tank, with the size of the tap representing the magnitude of clearance. Elimination is first order, as the more water there is in the tank, the faster it will flow out of the "clearance" tap. Inter-compartmental clearance is represented by flow through pipes connecting the bottom of two tanks. The flow rate through the pipe representing the magnitude of the inter-compartmental clearance; increasing the diameter of the pipe increases the compartmental clearance.

The two hydraulic analogues below are a simplified illustration of the key kinetic differences between the opioids alfentanil and fentanyl. The latter has a deep, slowly equilibrating peripheral compartment that can lead to prolonged recovery from long intravenous infusions, as the decline in the post-infusion central compartment concentrations have a significant contribution from the "emptying" of the peripheral compartment in addition to clearance.



12.6 Example NONMEM Code for some Absorption Models

12.6.1 Data for extravascular dose only examples

File name should be 1comp_extraonly_data.csv

CID	TIME	ROUTE	AMT	DV	MDV
1	0	1	200	.	1
1	2.5	1	.	0.13309	0
1	5	1	.	0.27899	0
1	10	1	.	0.52404	0
1	15	1	.	0.50467	0
1	30	1	.	0.55787	0
1	60	1	.	0.48462	0
1	90	1	.	0.46952	0
1	120	1	.	0.37605	0
1	180	1	.	0.38274	0
1	240	1	.	0.31569	0
1	300	1	.	0.22836	0
1	360	1	.	0.25917	0
1	480	1	.	0.14387	0
1	600	1	.	0.10837	0
2	0	1	200	.	1
2	2.5	1	.	0.052278	0
2	5	1	.	0.1168	0
2	10	1	.	0.26816	0
2	15	1	.	0.22321	0
2	30	1	.	0.20094	0
2	60	1	.	0.26505	0
2	90	1	.	0.42739	0
2	120	1	.	0.25939	0
2	180	1	.	0.25454	0
2	240	1	.	0.16357	0
2	300	1	.	0.13905	0
2	360	1	.	0.1691	0
2	480	1	.	0.093536	0
2	600	1	.	0.076136	0

12.6.2 NONMEM Code - extravascular data only - first order absorption

```
$PROBLEM - 1 COMP SYSTEMIC MODEL WITH FIRST ORDER ABSORPTION

$INPUT ID TIME ROUTE AMT DV MDV

$DATA 1comp_extraonly_data.csv IGNORE=C

$SUBROUTINES  ADVAN2 TRANS2

$PK
   CL = THETA(1)*EXP(ETA(1))
V  = THETA(2)
   KA = THETA(3)
   ALAG1 = THETA(4)

   S2 = V

$THETA
       (0,1,)     ;POPCLoverF
       (0,100,)   ;POPVoverF
       (0,0.5,)   ;POPKA
       (0,0.5,)   ;POPALAG

$OMEGA
  0.04    ;CLPPV

$SIGMA
  0.01   ;RUVPROP

$ERROR
  Y = F*(1 + ERR(1))
  IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID  TIME AMT CL V KA ALAG1 CWRES IPRE NOPRINT ONEHEADER FILE=*.fit
```

12.6.3 NONMEM Code - extravascular data only - first order absorption with common ETA

```
$PROBLEM - 1 COMP SYSTEMIC MODEL WITH FIRST ORDER ABSORPTION

$INPUT ID TIME ROUTE AMT DV MDV

$DATA 1comp_extraonly_data.csv IGNORE=C

$SUBROUTINES  ADVAN2 TRANS2

$PK
   FVAR = EXP(ETA(1))  ;common intersubject variability in F
   CL = THETA(1)*FVAR*EXP(ETA(2))
```

```
V  = THETA(2)*FVAR
   KA = THETA(3)
   ALAG1 = THETA(4)

   S2 = V

$THETA
       (0,1,)     ;POPCLoverF
       (0,100,)   ;POPVoverF
       (0,0.5,)   ;POPKA
       (0,0.5,)   ;POPALAG

$OMEGA
  0.04    ;FVAR
  0.04    ;PPVCL

$SIGMA
  0.01   ;RUVPROP

$ERROR
  Y = F*(1 + ERR(1))
  IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID  TIME AMT CL V KA ALAG1 FVAR CWRES IPRE NOPRINT ONEHEADER FILE=*.fit
```

12.6.4 NONMEM Code - extravascular data only - 3 transit compartments

```
$PROBLEM - 1 COMP SYSTEMIC MODEL WITH THREE TRANSIT COMPARTMENTS FOR ABSORPTION

$INPUT ID TIME ROUTE AMT DV MDV

$DATA 1comp_extraonly_data.csv IGNORE=C

$SUBROUTINES  ADVAN13 TOL=9

$MODEL
COMP=(ABS, DEFDOS)
COMP=(BLOOD, DEFOBS)
COMP=(TRANSIT1)
COMP=(TRANSIT2)
COMP=(TRANSIT3)

$PK
CL  = THETA(1)*EXP(ETA(1))
   V   = THETA(2)
   KTR = THETA(3)

   S2 = V

$DES
    C =  A(2)/V ; Central compartment concentration
```

```
;Absorption compartment
DADT(1) =  -KTR*A(1)
;Systemic model
DADT(2) =   KTR*A(5) - CL*C

;Transit compartments
DADT(3) = KTR*A(1) - KTR*A(3)
DADT(4) = KTR*A(3) - KTR*A(4)
DADT(5) = KTR*A(4) - KTR*A(5)
;add or remove transit compartments here

$THETA
       (0,1,)     ;POPCLoverF
       (0,100,)   ;POPVoverF
       (0,0.5,)   ;POPKTR

$OMEGA
  0.04    ;CLPPV

$SIGMA
  0.01   ;RUVPROP

$ERROR
  A1 = A(1)
  A2 = A(2)
  A3 = A(3)
  A4 = A(4)
  A5 = A(5)

  Y = F*(1 + ERR(1))
  IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID  TIME AMT A1 A2 A3 A4 A5 CL V KTR CWRES IPRE NOPRINT ONEHEADER FILE=*.fit
```

12.6.5 Data for intravenous and extravascular dose example

File name should be 1comp_extraiv_data.csv

CID	TIME	ROUTE	AMT	DV	MDV	CMT
1	0	0	200	.	1	2
1	2.5	0	.	4.2476	0	2
1	5	0	.	3.7739	0	2
1	10	0	.	2.2433	0	2
1	15	0	.	1.4836	0	2
1	30	0	.	1.1317	0	2
1	60	0	.	0.53769	0	2
1	90	0	.	0.83701	0	2
1	120	0	.	0.42357	0	2
1	180	0	.	0.47611	0	2
1	240	0	.	0.54053	0	2
1	300	0	.	0.22593	0	2
1	360	0	.	0.45987	0	2
1	480	0	.	0.35642	0	2
1	600	0	.	0.27012	0	2
2	0	1	200	.	1	1
2	2.5	1	.	0.13309	0	2
2	5	1	.	0.27899	0	2
2	10	1	.	0.52404	0	2
2	15	1	.	0.50467	0	2
2	30	1	.	0.55787	0	2
2	60	1	.	0.48462	0	2
2	90	1	.	0.46952	0	2
2	120	1	.	0.37605	0	2
2	180	1	.	0.38274	0	2
2	240	1	.	0.31569	0	2
2	300	1	.	0.22836	0	2
2	360	1	.	0.25917	0	2
2	480	1	.	0.14387	0	2
2	600	1	.	0.10837	0	2
3	0	0	200	.	1	2
3	2.5	0	.	1.3167	0	2
3	5	0	.	1.6598	0	2
3	10	0	.	1.4208	0	2
3	15	0	.	1.0973	0	2
3	30	0	.	0.58057	0	2
3	60	0	.	0.38369	0	2
3	90	0	.	0.30652	0	2
3	120	0	.	0.29308	0	2

3	180	0	.	0.26668	0	2
3	240	0	.	0.2523	0	2
3	300	0	.	0.17382	0	2
3	360	0	.	0.1754	0	2
3	480	0	.	0.19114	0	2
3	600	0	.	0.10163	0	2
4	0	1	200	.	1	1
4	2.5	1	.	0.05878	0	2
4	5	1	.	0.11252	0	2
4	10	1	.	0.24219	0	2
4	15	1	.	0.41246	0	2
4	30	1	.	0.36418	0	2
4	60	1	.	0.28775	0	2
4	90	1	.	0.33787	0	2
4	120	1	.	0.21029	0	2
4	180	1	.	0.18628	0	2
4	240	1	.	0.15313	0	2
4	300	1	.	0.17752	0	2
4	360	1	.	0.18525	0	2
4	480	1	.	0.12491	0	2
4	600	1	.	0.091946	0	2

12.6.6 NONMEM Code - intravenous and extravascular doses - first order absorption

```
$PROBLEM - 1 COMP SYSTEMIC MODEL IV DATA and EXTRAVASCULAR DATA

$INPUT ID TIME ROUTE AMT DV MDV CMT
;the CMT data item makes sure the dose goes to the correct compartment - 1 for extravascular dose, 2 for i.v. dose

$DATA 1comp_extraiv_data.csv IGNORE=C

$SUBROUTINES  ADVAN2 TRANS2

$PK
   CL = THETA(1)*EXP(ETA(1))
V  = THETA(2)

   IF (ROUTE.EQ.1) THEN  ;extravascular parameters
    FLGT = THETA(3) + ETA(2)   ;only use additive error here to allow negative values
    F1 = EXP(FLGT)/(1+EXP(FLGT))   ;logit transform to keep F between 0 and 1
    KA = THETA(4)
    ALAG1 = THETA(5)
   ELSE
    F1 = 1
;others will be set to zero unless a value is specified here
   ENDIF

   S2 = V

$THETA
       (0,1,)       ;POPCL
       (0,100,)     ;POPV
       (-12,0.5,12) ;POPFLGT  ;can range between approx. -12 (F near 0) and + 12 (F near 1)
       (0,0.5,)     ;POPKA
       (0,0.5,)     ;POPALAG

$OMEGA
  0.04    ;CLPPV
  0.04    ;FLGTPPV

$SIGMA
  0.01   ;RUVPROP

$ERROR
  Y = F*(1 + ERR(1))
  IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID  TIME ROUTE AMT CL V KA ALAG1 F1 FLGT CWRES IPRE NOPRINT ONEHEADER FILE=*.fit
```

12.7 Between Occasion Variability

12.7.1 Data for between occasion variability example

File name should be 1comp_iv_bov.csv

CID	OCC	TIME	AMT	DV	MDV	EVID
1	1	0	200	.	1	1
1	1	2.5	.	1.3167	0	0
1	1	5	.	1.6598	0	0
1	1	10	.	1.4208	0	0
1	1	15	.	1.0973	0	0
1	1	30	.	0.58057	0	0
1	1	60	.	0.38369	0	0
1	1	90	.	0.30652	0	0
1	1	120	.	0.29308	0	0
1	1	180	.	0.26668	0	0
1	1	240	.	0.2523	0	0
1	1	300	.	0.17382	0	0
1	1	360	.	0.1754	0	0
1	1	480	.	0.19114	0	0
1	1	600	.	0.10163	0	0
1	2	0	200	.	1	4
1	2	2.5	.	2.8322	0	0
1	2	5	.	3.0643	0	0
1	2	10	.	1.7029	0	0
1	2	15	.	1.0691	0	0
1	2	30	.	0.58435	0	0
1	2	60	.	0.37552	0	0
1	2	90	.	0.30784	0	0
1	2	120	.	0.33448	0	0
1	2	180	.	0.34844	0	0
1	2	240	.	0.30801	0	0
1	2	300	.	0.26821	0	0
1	2	360	.	0.23287	0	0
1	2	480	.	0.21906	0	0
1	2	600	.	0.13808	0	0
2	1	0	200	.	1	1
2	1	2.5	.	4.2476	0	0
2	1	5	.	3.7739	0	0
2	1	10	.	2.2433	0	0
2	1	15	.	1.4836	0	0
2	1	30	.	1.1317	0	0
2	1	60	.	0.53769	0	0

2	1	90	.	0.83701	0	0
2	1	120	.	0.42357	0	0
2	1	180	.	0.47611	0	0
2	1	240	.	0.54053	0	0
2	1	300	.	0.22593	0	0
2	1	360	.	0.45987	0	0
2	1	480	.	0.35642	0	0
2	1	600	.	0.27012	0	0
2	2	0	200	.	1	4
2	2	2.5	.	4.5892	0	0
2	2	5	.	3.5298	0	0
2	2	10	.	1.9632	0	0
2	2	15	.	0.92577	0	0
2	2	30	.	0.59046	0	0
2	2	60	.	0.62927	0	0
2	2	90	.	0.54218	0	0
2	2	120	.	0.53718	0	0
2	2	180	.	0.30519	0	0
2	2	240	.	0.30205	0	0
2	2	300	.	0.24115	0	0
2	2	360	.	0.25648	0	0
2	2	480	.	0.18967	0	0
2	2	600	.	0.20485	0	0

12.7.2 Example NONMEM Code for Between Occasion Variability

```
$PROBLEM - 1 COMP SYSTEMIC MODEL IV DOSE, TWO OCCASIONS PER SUBJECT

$INPUT ID OCC TIME AMT DV MDV EVID
;OCC flag identifies dose occasion within a subject
;EVID=4 resets the kinetic system so that time can begin at zero for the second occasion in each subject

$DATA 1comp_iv_bov.csv IGNORE=C

$SUBROUTINES  ADVAN1 TRANS2

$PK
    BSV = ETA(1)
    IF (OCC.EQ.1) BOV=ETA(2)
    IF (OCC.EQ.2) BOV=ETA(3)

   CL = THETA(1)*EXP(BSV+BOV)
V  = THETA(2)

   S1 = V

$THETA
       (0,0.5,)   ;POPCL
       (0,10,)    ;POPV

$OMEGA
  0.04    ;CLBSV
$OMEGA BLOCK(1)
  0.02    ;BOV1
$OMEGA BLOCK(1) SAME ;BOV2

$SIGMA
  0.01   ;RUVPROP

$ERROR
  Y = F*(1 + ERR(1))
  IPRE=F

$ESTIMATION METHOD=1 INTERACTION MAXEVALS=9999 POSTHOC NOABORT NSIG=3 SIGL=9

$COVARIANCE  UNCONDITIONAL SLOW SIGL=12 PRINT=E

$TABLE ID  TIME AMT CL V ETA1 ETA2 ETA3 CWRES IPRE NOPRINT ONEHEADER FILE=*.fit
```

12.8 References (Online Supplement)

1. R Core Team. R: A language and environment for statistical computing. R Foundation for Statistical Computing, Vienna, Austria. ISBN 3-900051-07-0, URL http://www.R-project.org/ (2012)

2. Icon Development Solutions, Dublin, Ireland. NONMEM Software. URL http://www.iconplc.com/technology/products/nonmem/ (2012)

3. Holford, N.H.G. Wings for NONMEM. URL http://wfn.sourceforge.net/ (2011)

4. Beal, S.L. Ways to fit a PK model with some data below the quantification limit. *J. Pharmacokinet. Pharmacodyn.***28,**481-504 (2001).

5. Upton, R.N., Foster, D.J., Christrup, L.L., Dale, O., Moksnes, K., Popper, L.A physiologically-based recirculatory meta-model for nasal fentanyl in man. *J. Pharmacokinet. Pharmacodyn.***39,**561-76 (2012).

6. Capper, S.J, Loo, S., Geue, J.P., Upton, R.N.,Ong, J., Macintyre, P.E., Ludbrook, G.L. Pharmacokinetics of fentanyl after subcutaneous administration in volunteers. *Eur. J.Anaesth.***27,**241-6 (2010).

7. Wang, Y. Derivation of various NONMEM estimation methods. *J Pharmacokinet. Pharmacodyn.***34**, 575-93 (2007).

8. Bolker, B.M. *Ecological models and Data in R.* (Princeton University Press, Princeton, 2008).

9. Upton, R.N. Calculating the hybrid (macro) rate constants of a three-compartment mamillary pharmacokinetic model from known micro-rate constants. *J. Pharmacol. Toxicol Methods*. **49,** 65-8 (2004).